Find all lines on cubic surface Problem: Find the singular point of the cubic surface
$$ZX^2 - Y^3 + TZ^2 = 0.$$
And also determine all the lines on the surface.
My attempt: I calculated the singular point by setting the partial derivatives to $0$ and I got $P = (0 : a : 0 : b)$ where $a, b$ are not both zero. Did I get this right?
About calculating the lines, I have no idea where to start. Is there perhaps some sort of technique/method/proposition for doing this? Can someone perhaps show me as an answer how to do this for this problem since I have two more of this type to do and then I can practice on those other two.
As a side question: Is there perhaps some criterion of determining the amount of total lines on the surface?
Note: We are in $\mathbb{P}^{3}$ over an algebraically closed field which does not have characteristic $2$.
 A: a) It is quite easy to see that the only singularity of the projective surface $S\subset \mathbb P^3$ given by the equation $ZX^2 - Y^3 + TZ^2 = 0$ is the point $$P=[x:y:z:t]=[0:0:0:1]\in S$$
That type of singularity is called $E_6$ by the specialists.
b) It is also quite clear that the line $L$ given by the two equations $Y=Z=0$ lies on $S$ (i.e. $L\subset S$).
c) What is true but not clear at all however is that $L$ is the only line of $\mathbb P^3$ lying on $S$.
I know no elementary proof of that fact, which follows from the rather complicated classification of  singular cubic surfaces in projective $3$-space.
Reference
Bruce-Wall's article On the Classification of Cubic Surfaces, Lemma 4, page 251  and second table, page 255.
(Beware that their paper is written in a rather old-fashioned  style.)
A: Let me give a computational answer. The idea is to form the general line
$$x = p_1 + l q_1, \quad y = p_2 + l q_2, \quad z = p_3 + l q_3, \quad
t = p_4 + l q_4$$
Substitute that into the cubic $f(x,y,z,t) = z x^2 - y^3 + t z^2$, and collect coefficients $a_i$ of $l$. Form an Ideal $I$ from these coefficients $a_i$.
It is an ideal in $\mathbb{Q}[p_i,q_j]$.
Now consider the 6 Plücker coordinates
$$\xi_i : w_i - \psi_i(p_k,q_l)$$
where the $\psi_i$ are the six $2$-minors of
$$\begin{pmatrix} p_1 & p_2 & p_3 & p_4 \\ q_1 & q_2 & q_3 & q_4 \end{pmatrix}$$
Form a ring $T=\mathbb{Q}[p_i, q_j, w_l]$ and take the ideal $J = I + (\xi_i)_i$ of $T$. Eliminate from this ideal the variables $p_i,q_j$, it remains an Ideal $L$ of $T_1=\mathbb{Q}[w_1,\ldots,w_6]$. The zeros of this ideal $L$ are the points in Plücker space corresponding to the lines on $V(f)$. Now to see that only one line exist, do a generic linear projection from
$$\mathbf{P}^5 = \mathrm{proj}( T_1) \to \mathbf{P}^1 = \mathrm{proj} (\mathbb{Q}[u_1,u_2])$$
Compute the ideal of the image by elimination and find out, that it is
$$(a u_1 + b u_2)^{17}$$
with $a,b \in \mathbb{Z}$ depending on the projection choosen. Why $17$ and not $27$? (I don't know).
On Macaulay2 homepage
https://faculty.math.illinois.edu/Macaulay2/
you find a web interface of Macaulay 2 on which you can run the code below (computation time is short):
https://www.unimelb-macaulay2.cloud.edu.au/#home
(You can copy the code below with Control-C and insert it with Control-V on line i1 in the web interface.)
-- We need all these variables for our computation

R = QQ[p_1..p_4,q_1..q_4,l,x,y,z,t]


-- our singular cubic

f = z * x^2 - y^3 + t * z^2

-- as a test a nonsingular cubic
-- f = x^3 + y^3 + z^3 + t^3

-- the variables of the projective Space P^3

vs = {x,y,z,t}

-- we make a substitute list x => p_1 + l * q_1, y => p_2 + l * q_2, and so on

sl = toList apply(1..4, i-> vs_(i-1) => p_i + l * q_i)

-- we apply these substitute list on f getting a polynomial in p_1..q_4 and l
-- we want that it vanishes exactly in l, so we take coefficients with respect to l

cfs = coefficients(sub(f, sl), Variables=>{l})

-- we form the ideal of these coefficients, it is an ideal of QQ[p_1..p_4,q_1..q_4]

iJ = ideal cfs_1

-- we provide explicitly the ring S with only p_i, q_i as variables

S = QQ[p_1..q_4]

-- now we prepare the canonical map from R to S which is identity on p_i, q_j and 0 otherwise

psi=map(S,R)

-- now we transfer our ideal in this smaller ring

iJ1 = psi(iJ)

-- we switch the default input to ring S

use S

-- we prepare to map the solutions of iJ to Plücker-coordinates

mM = matrix {{p_1,p_2,p_3,p_4},{q_1,q_2,q_3,q_4}}

-- the 6 polynomials in iK are the expressions of the Plücker variables

iK = minors(2, mM)

-- w_1..w_6 are our Plücker-coordinates, T has p_i, q_j, w_s as coordinates

T = S[w_1..w_6]

gamma = map(T,S)

-- we form the ideal w_a = psi_a(p_i, q_j) where psi_a are the Plücker expressions

iK1 = ideal toList(apply(1..6, i->w_i - gamma(iK_(i-1))))

-- our goal is to eliminate the p_i, q_j from the ideal made of iJ and iK1
-- we call this ideal iK2

iK2 = iK1 + gamma(iJ1)

-- following line can be ignored
elv = apply({p_1,p_2,p_3,p_4,q_1,q_2,q_3,q_4}, z->gamma(z))

-- we do the elimination by kernel computation

-- T1 is our final coordinate space

T1 = QQ[w_1..w_6]

-- the following two lines do elimination of p_i, q_j by a kernel computation.

rho=map(T/iK2, T1)

-- iL is our solution ideal in Plücker coordinates. It is of dim = 1 (projective dim = 0)
-- and degree 27

iL = ker rho

-- just to get a bit better informed, we calculate the primaryDecomposition of iL

pd1 = primaryDecomposition iL

iL1 = pd1_0

-- we want to project V(iL) to a line with a random linear projection

uw1 = sum toList(apply(1..6, i->random(-20,20) * w_i))

uw2 = sum toList(apply(1..6, i->random(-20,20) * w_i))

-- our ring that corresponds to a projective line

T2 = QQ[u_1,u_2]

-- here we do the projection by elimination

rho1 = map(T1/iL, T2, {uw1,uw2})

-- iL2 contains only one element g

iL2 = ker rho1

g = iL2_0

-- see for yourself, that g1 = l^17 with l linear in u_1,u_2
-- why 17 and not 27: I do not know, perhaps a question for Georges Elencwajg

g1 = factor g

A: Here is a direct way of showing the cubic surface $S=\{ZX^2-Y^3+Z^2T=0\}$ with an $E_6$-sigularity contains a single line.
Suppose there is another line $L$ in $S$ and is different from the line $\{Y=Z=0\}$ through the singularity, then $L$ must contain a point $p$ of the form $$p=(x_0,y_0,1,t_0),$$
and $L$ lies in the tangent hyperplane $T_pS$. The condition $p\in S$ requires that $t_0=y_0^3-x_0^2$, so $T_pS$ has equation
$$2x_0X-3y_0^2Y+(2y_0^3-x_0^2)Z+T=0.$$
The intersection $C=T_pS\cap S$ is a cubic plane curve and $C$ contains $L$ as an irreducible component, therefore $C$ is reducible. However, by isolating $T$ in the linear equation and cancelling $T$ in the cubic surface equation, one finds the equation of $C$ is
$$ZX^2-Y^3-Z^2[2x_0X-3y_0^2Y+(2y_0^3-x_0^2)Z]$$
in the projective plane with coordinate $X,Y,Z$. Of course, $C$ is singular at $p=(x_0,y_0,1)$ since it is tangency point. However, a direct computation of partial derivatives shows that $p$ is the only singularity of $C$ and it is an ordinary double point if $y_0\neq 0$, is an ordinary cusp if $y_0=0$. Therefore $C$ is irreducible, which is a contradiction.
